
	Reason for change:
	In function f_CharPacking_IA5toGsm7Bit_WithFiller the 3rd parameter p_NumFillbits is used to to add a filler bit at the end of the header and in front of the actual data 

According to 23.040 sec 9.2.3.4

If 7 bit data is used and the TP‑UD‑Header does not finish on a septet boundary then fill bits are inserted after the last Information Element Data octet up to the next septet boundary so that there is an integral number of septets for the entire TP‑UD header

In the current TTCN implementation the filler bit is added to the bitstring from the return octet of f_CharPacking_IA5toGsm7Bit, this is incorrect as the filler bit needs to be added at the start of the SMS message so the first character starts at the septets boundary .

	
	

	Summary of change:
	p_NumFillbits is passed as a parameter to f_CharPacking_IA5toGsm7Bit and the first character of the SMS starts on the septets boundary. “The” -> 546865 results in 1100101110100010101000

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc68623623]Corrections required
0.1 [bookmark: _Toc68623624]Correction to f_CharPacking_IA5toGsm7Bit_WithFiller
	Function name
	f_CharPacking_IA5toGsm7Bit_WithFiller

	Reason for change
	In function f_CharPacking_IA5toGsm7Bit_WithFiller the 3rd parameter p_NumFillbits is used to to add a filler bit at the end of the header and in front of the actual data 

According to 23.040 sec 9.2.3.4

If 7 bit data is used and the TP‑UD‑Header does not finish on a septet boundary then fill bits are inserted after the last Information Element Data octet up to the next septet boundary so that there is an integral number of septets for the entire TP‑UD header

In the current TTCN implementation the filler bit is added to the bitstring from the return octet of f_CharPacking_IA5toGsm7Bit. This is incorrect as the filler bit needs to be added at the start of the SMS message so the first character starts at the septets boundary.

	Summary of change
	v_Fillbits is updated and passed into function f_CharPacking_IA5toGsm7Bit.
Removed update to v_BitString at the end of the function as its incorrect to update it here.

	TTCN module
	IMS_SMS_Messages

	MCC160 Comment
	



Before Change:
	function f_CharPacking_IA5toGsm7Bit_WithFiller(charstring p_Text,
                                                 Gsm7BitPacking_Type p_Packing,
                                                 integer p_NumFillbits) return octetstring
  {  // @sic R5s201446 sic@
    var octetstring v_OctetString;
    var bitstring v_BitString := ''B;
    var bitstring v_Fillbits := ''B;
    var integer v_BitStringLength := 0;

    // conversion of the text string, padding is added
    v_OctetString := f_CharPacking_IA5toGsm7Bit(p_Text, p_Packing);
    // conversion to a bitstring
    v_BitString := oct2bit(v_OctetString);
    // note the length of the bitstring
    v_BitStringLength := lengthof(v_BitString);
    // calculate the wanted fillbits
    v_Fillbits := int2bit(0, p_NumFillbits);  // 1..6 fill bits, all 0 @sic R5s201329 sic@
    // concatenate fillbits and bitstring
    v_BitString := v_Fillbits & v_BitString;
    // cut the resulting bitstring to the length of the original bitstring - assumption that the user of this function knows there is enough padding
    v_BitString := substr(v_BitString, 0, v_BitStringLength);
    // convert resulting bitstring to octets
    return(bit2oct(v_BitString));
  }



After Change:
	  function f_CharPacking_IA5toGsm7Bit_WithFiller(charstring p_Text,
                                                 Gsm7BitPacking_Type p_Packing,
                                                 integer p_NumFillbits) return octetstring
  {  // @sic R5s201446 sic@
    var octetstring v_OctetString;
    var bitstring v_BitString := ''B;
    var bitstring v_Fillbits := ''B;
    var integer v_BitStringLength := 0;
    v_Fillbits := int2bit(0, p_NumFillbits);  // 1..6 fill bits, all 0 @sic R5s201329 sic@
    // conversion of the text string, padding is added
    v_OctetString := f_CharPacking_IA5toGsm7Bit(p_Text, p_Packing, v_Fillbits);
    // conversion to a bitstring
    v_BitString := oct2bit(v_OctetString);
    // note the length of the bitstring
    v_BitStringLength := lengthof(v_BitString);
    // calculate the wanted fillbits
    // concatenate fillbits and bitstring
  //  v_BitString := v_Fillbits & v_BitString; not needed here,as filler bits are done within f_CharPacking_IA5toGsm7Bit
      // cut the resulting bitstring to the length of the original bitstring - assumption that the user of this function knows there is enough padding
    v_BitString := substr(v_BitString, 0, v_BitStringLength);
    // convert resulting bitstring to octets
    return(bit2oct(v_BitString));
  }



0.2 [bookmark: _Toc68623625]Correction to f_CharPacking_IA5toGsm7Bit
	Function name
	f_CharPacking_IA5toGsm7Bit

	Reason for change
	In function f_CharPacking_IA5toGsm7Bit_WithFiller the 3rd parameter p_NumFillbits is used to to add a filler bit at the end of the header and in front of the actual data 

According to 23.040 sec 9.2.3.4

If 7 bit data is used and the TP‑UD‑Header does not finish on a septet boundary then fill bits are inserted after the last Information Element Data octet up to the next septet boundary so that there is an integral number of septets for the entire TP‑UD header

In the current TTCN implementation the filler bit is added to the bitstring from the return octet of f_CharPacking_IA5toGsm7Bit. This is incorrect as the filler bit needs to be added at the start of the SMS message so the first character starts at the septets boundary.

	Summary of change
	Added new parameter to receive filler bits and updated v_BitString and v_BitStringLength

	TTCN module
	CommonDefs

	MCC160 Comment
	



Before Change:
	function f_CharPacking_IA5toGsm7Bit(charstring p_Text,
                                      Gsm7BitPacking_Type p_Packing) return octetstring
  { /* @sic R5s130103 change 3: consideration of non alpha numeric characters sic@ */
    var octetstring v_OctetString := char2oct(p_Text);   // "Hallo" -> '48616C6C6F'O
    var bitstring v_BitString := ''B;
    var bitstring v_Padding := ''B;
    var integer v_TextLength;
    var integer v_BitStringLength := 0;
    var integer v_IntValue;
    var integer v_NoOfOctets;
    var integer v_NoOfPaddingBits;
    var integer v_Index;
    var integer i;
    
    // convert ascii table to GSM 7 bit table
    v_OctetString := fl_XlatAcsiiToGsm7Bit(v_OctetString);                 //  @sic R5s130103 change 3 sic@
    v_TextLength := lengthof(v_OctetString);

    if (((p_Packing == SMS_Packing) and (v_TextLength > 160)) or           // a maximum of 160 characters for SMS  (i.e. 140 octets)
        ((p_Packing == CBS_Packing) and (v_TextLength > 93)) or            // a maximum of  93 characters for CBS  (i.e.  82 octets)
        ((p_Packing == USSD_Packing) and (v_TextLength > 182))) {          // a maximum of 182 characters for USSD (i.e. 160 octets)
      FatalError(__FILE__, __LINE__, "input string is too long");
    }

    for (i:=0; i<v_TextLength; i:=i+1) {
      v_IntValue := oct2int(v_OctetString[i]);
      v_BitString := int2bit(v_IntValue, 7) & v_BitString;  // concat 7 bits only
      v_BitStringLength := v_BitStringLength + 7;
    }
    // add padding:
    v_NoOfOctets := (v_BitStringLength + 7) / 8;
    v_NoOfPaddingBits := (v_NoOfOctets * 8) - v_BitStringLength;
    if (v_NoOfPaddingBits > 0) {
      if ((p_Packing == USSD_Packing) and (v_NoOfPaddingBits == 7)) {
        v_Padding := '0001101'B;                     // use <CR> instead of the @ character; see TS 23.038 cl. 6.1.2.3.1
      } else {
        v_Padding := int2bit(0, v_NoOfPaddingBits);  // 1..7 padding bits, all 0
      }
    }
    v_BitString := v_Padding & v_BitString;

    // convert bitstring back to octetstring: the first character is at the end of the bitstring
    // => we need to take the octets from the end of the bitstring
    v_OctetString := ''O;
    for (i:=0; i < v_NoOfOctets; i:=i+1) {
      v_Index := (v_NoOfOctets - (i+1)) * 8;    // get start index of the next octet
      v_OctetString[i] := bit2oct(substr(v_BitString, v_Index, 8));
    }
    return v_OctetString;
  }



After Change:
	function f_CharPacking_IA5toGsm7Bit(charstring p_Text,
                                      Gsm7BitPacking_Type p_Packing, 
                                      bitstring p_fillBits := ''B) return octetstring //WA# filler bits added.
  { /* @sic R5s130103 change 3: consideration of non alpha numeric characters sic@ */
    var octetstring v_OctetString := char2oct(p_Text);   // "Hallo" -> '48616C6C6F'O
    
    var bitstring v_BitString := p_fillBits; // filler bits added.
    var bitstring v_Padding := ''B;
    var integer v_TextLength;
    var integer v_BitStringLength := lengthof(p_fillBits); //WA# take the length of the filler bits into consideration.
    var integer v_IntValue;
    var integer v_NoOfOctets;
    var integer v_NoOfPaddingBits;
    var integer v_Index;
    var integer i;
    
    // convert ascii table to GSM 7 bit table
    v_OctetString := fl_XlatAcsiiToGsm7Bit(v_OctetString);                 //  @sic R5s130103 change 3 sic@
    v_TextLength := lengthof(v_OctetString);

    if (((p_Packing == SMS_Packing) and (v_TextLength > 160)) or           // a maximum of 160 characters for SMS  (i.e. 140 octets)
        ((p_Packing == CBS_Packing) and (v_TextLength > 93)) or            // a maximum of  93 characters for CBS  (i.e.  82 octets)
        ((p_Packing == USSD_Packing) and (v_TextLength > 182))) {          // a maximum of 182 characters for USSD (i.e. 160 octets)
      FatalError(__FILE__, __LINE__, "input string is too long");
    }

    for (i:=0; i<v_TextLength; i:=i+1) {
      v_IntValue := oct2int(v_OctetString[i]);
      v_BitString := int2bit(v_IntValue, 7) & v_BitString;  // concat 7 bits only
      v_BitStringLength := v_BitStringLength + 7;
    }
    // add padding:
    v_NoOfOctets := (v_BitStringLength + 7) / 8;
    v_NoOfPaddingBits := (v_NoOfOctets * 8) - v_BitStringLength;
    if (v_NoOfPaddingBits > 0) {
      if ((p_Packing == USSD_Packing) and (v_NoOfPaddingBits == 7)) {
        v_Padding := '0001101'B;                     // use <CR> instead of the @ character; see TS 23.038 cl. 6.1.2.3.1
      } else {
        v_Padding := int2bit(0, v_NoOfPaddingBits);  // 1..7 padding bits, all 0
      }
    }
    v_BitString := v_Padding & v_BitString;

    // convert bitstring back to octetstring: the first character is at the end of the bitstring
    // => we need to take the octets from the end of the bitstring
    v_OctetString := ''O;
    for (i:=0; i < v_NoOfOctets; i:=i+1) {
      v_Index := (v_NoOfOctets - (i+1)) * 8;    // get start index of the next octet
      v_OctetString[i] := bit2oct(substr(v_BitString, v_Index, 8));
    }
    return v_OctetString;
  }




